Example 8.11 Diffusion in a Slab with Nonhomogeneous Flux Boundary Condtions during the Charging of a Battery
Equation (8.1.27) is solved in Maple and the results are given below:

```
> restart:with(inttrans):with(plots):
> eq:=diff(u(x,t),t)=diff(u(x,t),x$2);
```

$$eq := \frac{\partial}{\partial t} u(x,t) = \frac{\partial^2}{\partial x^2} u(x,t) \tag{1}$$

```
> u(x,0):=0;
```

$$u(x,0) := 0 \tag{2}$$

```
> bc1:=diff(u(x,t),x)=0;
```

$$bc1 := \frac{\partial}{\partial x} u(x,t) = 0 \tag{3}$$

```
> bc2:=diff(u(x,t),x)=delta;
```

$$bc2 := \frac{\partial}{\partial x} u(x,t) = \delta \tag{4}$$

```
> eqs:=laplace(eq,t,s):
> eqs:=subs(laplace(u(x,t),t,s)=U(x),eqs);
```

$$eqs := s\, U(x) = \frac{d^2}{dx^2} U(x) \tag{5}$$

```
> bc1:=laplace(bc1,t,s):
> bc1:=subs(laplace(u(x,t),t,s)=U(x),bc1);
```

$$bc1 := \frac{d}{dx} U(x) = 0 \tag{6}$$

```
> bc2:=laplace(bc2,t,s):
> bc2:=subs(laplace(u(x,t),t,s)=U(x),bc2);
```

$$bc2 := \frac{d}{dx} U(x) = \frac{\delta}{s} \tag{7}$$

```
> dsolve(eqs,U(x));
```

$$U(x) = \_C1\, e^{\sqrt{s}\, x} + \_C2\, e^{-\sqrt{s}\, x} \tag{8}$$

```
> U(x):=c[1]*cosh(s^(1/2)*x)+c[2]*sinh(s^(1/2)*x);
```

$$U(x) := c_1 \cosh\left(\sqrt{s}\, x\right) + c_2 \sinh\left(\sqrt{s}\, x\right) \tag{9}$$

```
> eq0:=eval(subs(x=0,bc1)):
> eq1:=eval(subs(x=1,bc2)):
> con:=solve({eq0,eq1},{c[1],c[2]}):
> U(x):=subs(con,U(x));
```

$$U(x) := \frac{\delta \cosh\left(\sqrt{s}\, x\right)}{\sinh\left(\sqrt{s}\right) s^{3/2}} \tag{10}$$

```
> U(x):=factor(combine(simplify(U(x))));
```

$$U(x) := \frac{\delta \cosh(\sqrt{s}\, x)}{\sinh(\sqrt{s})\, s^{3/2}} \tag{11}$$

```
> P(s):=numer(U(x));
```

$$P(s) := \delta \cosh(\sqrt{s}\, x) \tag{12}$$

```
> Q(s):=denom(U(x));
```

$$Q(s) := \sinh(\sqrt{s})\, s^{3/2} \tag{13}$$

```
> solve(Q(s),s);
```

$$0 \tag{14}$$

```
> _EnvAllSolutions := true;
```

$$\mathit{\_EnvAllSolutions} := \mathit{true} \tag{15}$$

```
> solve(Q(s),s);
```

$$-\pi^2\, \_Z1\text{~}^2,\ 0 \tag{16}$$

```
> 0,0,-n^2*Pi^2;
```

$$0,\ 0,\ -n^2\,\pi^2 \tag{17}$$

```
> mu0:=0;
```

$$\mu 0 := 0 \tag{18}$$

```
> b[2]:=(s-mu0)^2*P(s)/Q(s);
```

$$b_2 := \frac{\sqrt{s}\,\delta \cosh(\sqrt{s}\, x)}{\sinh(\sqrt{s})} \tag{19}$$

```
> B[2]:=limit(b[2],s=0);
```

$$B_2 := \delta \tag{20}$$

```
> b[1]:=diff(b[2],s):
> B[1]:=limit(b[1],s=0);
```

$$B_1 := \frac{1}{2}\,\delta x^2 - \frac{1}{6}\,\delta \tag{21}$$

```
> A(s):=P(s)/diff(Q(s),s):
> A[n]:=simplify(subs(s=mu,A(s)));
```

$$A_n := \frac{2\,\delta \cosh(\sqrt{\mu}\, x)}{\cosh(\sqrt{\mu})\,\mu + 3\sinh(\sqrt{\mu})\,\sqrt{\mu}} \tag{22}$$

```
> A[n]:=simplify(subs(mu^(1/2)=I*n*Pi,mu=-n^2*Pi^2,A[n])):
> vars:={cos(n*Pi)=(-1)^n,sin(n*Pi)=0};
```

$$\mathit{vars} := \{\cos(n\,\pi) = (-1)^n,\ \sin(n\,\pi) = 0\} \tag{23}$$

```
> A[n]:=simplify(subs(vars,A[n])):
```

```
> A[n]:=simplify(subs(vars,expand(A[n])));
```

$$A_n := \frac{2\,(-1)^{1-n}\,\delta\cos(n\,\pi\,x)}{n^2\,\pi^2} \tag{24}$$

```
> b1s:=B[1]*subs(mu0=0,1/(s-mu0));
```

$$b1s := \frac{\frac{1}{2}\,\delta\,x^2 - \frac{1}{6}\,\delta}{s} \tag{25}$$

```
> b1t:=invlaplace(b1s,s,t);
```

$$b1t := \frac{1}{6}\,\delta\,(3\,x^2 - 1) \tag{26}$$

```
> b2s:=B[2]*subs(mu0=0,1/(s-mu0)^2);
```

$$b2s := \frac{\delta}{s^2} \tag{27}$$

```
> b2t:=invlaplace(b2s,s,t);
```

$$b2t := \delta\,t \tag{28}$$

```
> uns:=A[n]/(s-mu);
```

$$uns := \frac{2\,(-1)^{1-n}\,\delta\cos(n\,\pi\,x)}{n^2\,\pi^2\,(s-\mu)} \tag{29}$$

```
> unt:=invlaplace(uns,s,t);
```

$$unt := -\frac{2\,(-1)^{-n}\,\delta\cos(n\,\pi\,x)\,e^{\mu\,t}}{n^2\,\pi^2} \tag{30}$$

```
> unt:=subs(mu=-n^2*Pi^2,unt);
```

$$unt := -\frac{2\,(-1)^{-n}\,\delta\cos(n\,\pi\,x)\,e^{-n^2\,\pi^2\,t}}{n^2\,\pi^2} \tag{31}$$

```
> U:=b1t+b2t+Sum(unt,n=1..infinity);
```

$$U := \frac{1}{6}\,\delta\,(3\,x^2 - 1) + \delta\,t + \sum_{n=1}^{\infty}\left(-\frac{2\,(-1)^{-n}\,\delta\cos(n\,\pi\,x)\,e^{-n^2\,\pi^2\,t}}{n^2\,\pi^2}\right) \tag{32}$$

```
> u:=piecewise(t=0,0,t>0,subs(infinity=20,U)):
> plot3d(subs(delta=1,u),x=0..1,t=0..0.5,axes=boxed,title="Figure
  Exp. 8.20.",labels=[x,t,"u"],orientation=[-135,60]);
```

Figure Exp. 8.20.